

# Four Problems with Policy-Based Constraints and How to Fix Them

Dillan Mills  
Synopsys, Inc.  
Maricopa, AZ  
dillan@synopsys.com

Chip Haldane  
The Chip Abides, LLC  
Gilbert, AZ  
chip@thechipabides.com

## Abstract

This paper presents solutions to problems encountered in the implementation of policy classes for SystemVerilog constraint layering. Policy classes provide portable and reusable constraints that can be mixed and matched into the object being randomized. There have been many papers and presentations on policy classes since the original presentation by John Dickol at DVCon 2015. The paper addresses three problems shared by all public policy class implementations and presents a solution to a fourth problem. The proposed solutions introduce policy class inheritance, tightly pair policy definitions with the class they constrain, reduce the expense of defining common policies using macros, and demonstrate how to treat policies as disposable and lightweight objects. The paper concludes that the proposed solution improves the usability and efficiency of policy classes for SystemVerilog constraint layering.

## I. CONSTRAINTS AND POLICY CLASS REVIEW

When working with randomized objects in SystemVerilog, it is typically desired to control the randomization of the item being randomized. The common method of doing so is by using constrained random value generation. Constraints for the randomization of class properties can be provided within the class definition, which will require randomized properties to always pass the specified constraint check. For example, the following `addr_txn` class contains a constraint on the `addr` property to ensure it is always word aligned.

```
1 class addr_txn;  
2     rand bit [31:0] addr;  
3     rand int      size;  
4  
5     constraint c_size {size inside {1, 2, 4};}  
6 endclass
```

Fig. 1: A basic address transaction class

Constraints can also be provided outside the class definition at the point of randomization by using the `with` construct. This allows variation to the constraints based on the context of the randomization. For example, the same `addr_txn` class can have an instance of it randomized with the following constraint on its `data` property.

```
1 addr_txn txn = new();  
2 txn.randomize() with {addr == 32'hdeadbeef};
```

Fig. 2: An example of an inline constraint

Policy classes are a technique for SystemVerilog constraint layering with comparable performance to traditional constraint methods, while additionally providing portable and reusable constraints that can be mixed and matched into the object being randomized, originally presented by John Dickol [1] [2]. This is possible through SystemVerilog “Global Constraints”, which randomizes all lower-level objects whenever a top-level object is randomized, and solves all rand variables and constraints across the entire object hierarchy simultaneously. A summary of his approach can be found by examining the following code.

```
1 class policy_base#(type ITEM=uvm_object);  
2     ITEM item;
```

```

3
4     virtual function void set_item(ITEM item);
5         this.item = item;
6     endfunction
7 endclass
8
9 class policy_list#(type ITEM=uvm_object) extends policy_base#(ITEM);
10     rand policy_base#(ITEM) policy[$];
11
12     function void add(policy_base#(ITEM) pcy);
13         policy.push_back(pcy);
14     endfunction
15
16     function void set_item(ITEM item);
17         foreach(policy[i]) policy[i].set_item(item);
18     endfunction
19 endclass

```

Fig. 3: The **policy\_base** and **policy\_list** classes

These first two classes provide the base class types for the actual policies. This allows different, related policies to be grouped together in **policy\_list** and applied to the randomization call concurrently. Both of these classes are parameterized to a specific object type, so separate instances will be required for each class type that needs to be constrained.

```

1 class addr_policy_base extends policy_base#(addr_txn);
2     addr_range ranges[$];
3
4     function void add(addr_t min, addr_t max);
5         addr_range rng = new(min, max);
6         ranges.push_back(rng);
7     endfunction
8 endclass
9
10 class addr_permit_policy extends addr_policy_base;
11     rand int selection;
12
13     constraint c_addr_permit {
14         selection inside {[0:ranges.size()-1]};
15
16         foreach(ranges[i]) {
17             if(selection == i) {
18                 item.addr inside {[ranges[i].min:ranges[i].max - item.size]};
19             }
20         }
21     }
22 endclass
23
24 class addr_prohibit_policy extends addr_policy_base;
25     constraint c_addr_prohibit {
26         foreach(ranges[i]) {
27             !(item.addr inside {[ranges[i].min:ranges[i].max - item.size + 1]});
28         }
29     }
30 endclass

```

```

29     }
30 endclass

```

Fig. 4: Policies for constraining the `addr_txn` class

The next three classes contain implementations of the `policy_base` class for the `addr_txn` class. A list of valid address ranges can be stored in the `ranges` array. The `addr_permit_policy` will choose one of the ranges at random and constrain the address to be within the range, while the `addr_prohibit_policy` will exclude the address from randomizing inside any of the ranges in its list.

```

1  class addr_txn;
2      rand bit [31:0] addr;
3      rand int      size;
4      rand policy_base#(addr_txn) policy[$];
5
6      constraint c_size {size inside {1, 2, 4};}
7
8      function void pre_randomize;
9          foreach(policy[i]) policy[i].set_item(this);
10     endfunction
11 endclass
12
13 class addr_constrained_txn extends addr_txn;
14     function new;
15         addr_permit_policy permit = new;
16         addr_prohibit_policy prohibit = new;
17         policy_list#(addr_txn) pcy = new;
18
19         permit.add('h00000000, 'h0000FFFF);
20         permit.add('h10000000, 'h1FFFFFFF);
21         pcy.add(permit);
22
23         prohibit.add('h13000000, 'h130FFFFF);
24         pcy.add(prohibit);
25
26         this.policy = {pcy};
27     endfunction
28 endclass

```

Fig. 5: The `addr_txn` class and a constrained subclass

The final two classes show how these policies are used. The `addr_txn` class creates a `policy` queue, and during the `pre_randomize` function sets the handle to the `item` object within each policy to itself. The `addr_constrained_txn` subclass adds two policies to a local `pcy` list, one that permits the address to be within one of two ranges, and one that prohibits the address from being within a third range. The `addr_constrained_txn` class can then pass the local `pcy` list to the parent `policy` queue.

At this point, an instance of `addr_constrained_txn` can be created and randomized like normal, and the address will be constrained based on the policies provided.

```

1  addr_constrained_txn txn = new();
2  txn.randomize();

```

Fig. 6: Randomizing an instance of `addr_constrained_txn`

Further work on policy-based constraints has been presented since the original DVCon presentation in 2015. Kevin Vasconcellos and Jeff McNeal applied the concept to test configuration and added many nice utilities to the base policy class [3]. Chuck McClish extended the concept to manage real number values for User Defined Nettypes (UDN) and Unified Power Format (UPF) pins in an analog model [4]. Additionally, he defined a policy builder class that was used to generically build multiple types of policies while reducing repeated code that was shared between each policy class in the original implementation.

Although there has been extensive research on policies, this paper aims to address and provide solutions for three issues that have not been adequately resolved in previous implementations. Furthermore, a fourth problem that arose during testing of the upgraded policy package implementation will also be discussed and resolved.

## II. PROBLEM #1: PARAMETERIZED POLICIES

The first problem with the above policy implementation is that because policies are specialized using parameterization to the class they constrain, different specializations of parameterized classes cannot be grouped and indexed. This is most apparent when creating a class hierarchy. If you wanted to extend a class and add a new field to constrain then you need a new policy type. The new policy type requires an extra policy list to modify any new attributes in the child class and requires the user to know which layer of the class hierarchy defines each attribute they want to constrain. Imagine a complex class hierarchy with many layers of inheritance and extension, and new constrainable attributes on each layer (one common example is a multi-layered sequence API library, such as the example presented by Jeff Vance [5]). Using policies becomes cumbersome in this case because you need to manage a separate policy list for each level of the hierarchy and know what each list can and cannot contain. For example, extending the `addr_txn` class to create a version with parity checking results in a class hierarchy that looks like this:

```
1 class addr_txn;
2     rand bit [31:0] addr;
3     rand int      size;
4     rand policy_base#(addr_txn) addr_policy[$];
5
6     constraint c_size {size inside {1, 2, 4};}
7
8     function void pre_randomize;
9         foreach(addr_policy[i]) addr_policy[i].set_item(this);
10    endfunction
11 endclass
12
13 class addr_p_txn extends addr_txn;
14     rand bit parity;
15     rand policy_base#(addr_p_txn) addr_p_policy[$];
16
17     constraint c_parity {parity == $countones(addr) % 2;}
18
19     function void pre_randomize;
20         super.pre_randomize();
21         foreach(addr_p_policy[i]) addr_p_policy[i].set_item(this);
22     endfunction
23 endclass
24
25 class addr_constrained_txn extends addr_p_txn;
26     function new;
27         addr_permit_policy      permit = new;
28         addr_prohibit_policy    prohibit = new;
29         addr_parity_policy      parity = new;
30         policy_list#(addr_txn)  addr_pcy = new;
31         policy_list#(addr_p_txn) addr_p_pcy = new;
32
33         permit.add('h00000000, 'h0000FFFF);
```

```

34     permit.add('h10000000', 'h1FFFFFFF');
35     addr_pcy.add(permit);
36
37     prohibit.add('h13000000', 'h130FFFFFFF');
38     addr_pcy.add(prohibit);
39
40     parity.set(1'b1);
41     addr_p_pcy.add(parity);
42
43     this.addr_policy = {addr_pcy};
44     this.addr_p_policy = {addr_p_pcy};
45 endfunction
46 endclass

```

Fig. 7: Modified address transaction classes with a parity checking subclass and policies included

This class ends up with a lot of boilerplate code and is not very intuitive to use. The user needs to keep track of what the difference is between each policy list and apply constraints properly. Each additional subclass in a hierarchy will cause the constrained class to grow further. The solution to this problem is to replace the parameterized policy base classes with a combination of a policy interface class and a parameterized policy implementation class.

```

1  interface class policy;
2      pure virtual function void set_item(uvm_object item);
3  endclass
4
5  virtual class policy_imp#(type ITEM=uvm_object) implements policy;
6      protected rand ITEM m_item;
7
8      virtual function void set_item(uvm_object item);
9          if (!$cast(m_item, item)) begin
10              `uvm_warning("policy::set_item()", "Item/policy type mismatch")
11              this.m_item = null;
12              this.m_item.rand_mode(0);
13          end
14      endfunction: set_item
15  endclass: policy_imp
16
17  typedef policy policy_queue[$];

```

Fig. 8: The **policy** interface class and **policy\_imp** class

By abstracting the policy functionality into an interface class, all policy implementations under a common base class are given compatibility with each other and can be stored in a shared policy queue. The **uvm\_object** class is a good base class when using policies within UVM, as it will likely cover every class you will be randomizing. If working with another framework, a reasonable base class will need to be chosen. Because the **set\_item** function no longer receives a parameterized type input, the input needs to be cast to the implementation type. If this cast fails, the item is set to null and randomization is disabled.

```

1  class addr_policy extends policy_imp#(addr_txn);
2      addr_range ranges[$];
3
4      function void add(addr_t min, addr_t max);
5          addr_range rng = new(min, max);

```

```

6         ranges.push_back(rng);
7     endfunction
8 endclass
9
10    class addr_parity_policy extends policy_imp#(addr_p_txn);
11        protected bit parity;
12
13        constraint c_fixed_value {m_item != null -> m_item.parity == parity;}
14
15        function new(int parity);
16            this.parity = parity;
17        endfunction
18    endclass
19
20    class addr_permit_policy extends addr_policy;
21        rand int selection;
22
23        constraint c_addr_permit {
24            m_item != null -> (
25                selection inside {[0:ranges.size()-1]};
26
27                foreach(ranges[i]) {
28                    if(selection == i) {
29                        m_item.addr inside {[ranges[i].min:ranges[i].max - m_item.size]};
30                    }
31                }
32            )
33        }
34    endclass
35
36    class addr_prohibit_policy extends addr_policy;
37        constraint c_addr_prohibit {
38            m_item != null -> (
39                foreach(ranges[i]) {
40                    !(m_item.addr inside {[ranges[i].min:ranges[i].max - m_item.size + 1]});
41                }
42            )
43        }
44    endclass

```

Fig. 9: Address policies updated to use the new policy implementation

The address policies are updated to inherit from **policy\_imp** parameterized to the relevant transaction class. Additionally, the constraints are updated to verify the internal item handle is not null.

```

1    class addr_txn extends uvm_object;
2        rand bit [31:0]  addr;
3        rand int         size;
4        rand policy_queue policy;
5
6        constraint c_size {size inside {1, 2, 4};}
7
8        function void pre_randomize;
9            foreach(policy[i]) policy[i].set_item(this);

```

```

10     endfunction
11 endclass
12
13 class addr_p_txn extends addr_txn;
14     rand bit parity;
15
16     constraint c_parity {parity == $countones(addr) % 2;}
17 endclass
18
19 class addr_constrained_txn extends addr_p_txn;
20     function new;
21         policy_queue pcy;
22
23         addr_permit_policy permit = new();
24         addr_prohibit_policy prohibit = new();
25         addr_parity_policy fixed_p;
26
27         permit.add('h00000000, 'h0000FFFF);
28         permit.add('h10000000, 'h1FFFFFFF);
29         pcy.push_back(permit);
30
31         prohibit.add('h13000000, 'h130FFFFF);
32         pcy.push_back(prohibit);
33
34         fixed_p = new(1'b1);
35         pcy.push_back(fixed_p);
36
37         this.policy = {pcy};
38     endfunction
39 endclass

```

Fig. 10: Address transaction classes updated to use the new policy implementation

The new address transaction classes are simplified considerably. Only a single policy queue is required in the class hierarchy, and the constraint block does not need to keep track of any specialized lists. The **addr\_txn** class now subclasses **uvm\_object** to provide compatibility with the policy interface.

### III. PROBLEM #2: DEFINITION LOCATION

The second problem is “where do I define my policy classes?” This is not a complicated problem to solve, and most users likely place their policy classes in a file close to the class they are constraining. However, by explicitly pairing policy definitions with the class they constrain, users never need to hunt for the definitions or guess where they are located, and the definitions are given an appropriate and consistent home in the environment. This is done by embedding a **POLICIES** class inside the class being constrained. This class acts as a container for all relevant policies and allows policies to be created by simply referencing the embedded class through the **::** operator.

```

1 class addr_txn extends uvm_object;
2     rand bit [31:0] addr;
3     rand int size;
4     rand policy_queue policy;
5
6     constraint c_size {size inside {1, 2, 4};}
7
8     function void pre_randomize;
9         foreach(policy[i]) policy[i].set_item(this);

```

```

10     endfunction
11
12     class POLICIES;
13         class addr_policy extends policy_imp#(addr_txn);
14             // ... unchanged from previous example
15         endclass
16
17         class addr_permit_policy extends addr_policy;
18             // ... unchanged from previous example
19         endclass
20
21         class addr_prohibit_policy extends addr_policy;
22             // ... unchanged from previous example
23         endclass
24     endclass: POLICIES
25 endclass
26
27 class addr_p_txn extends addr_txn;
28     rand bit parity;
29
30     constraint c_parity {parity == $countones(addr) % 2;}
31
32     class POLICIES extends addr_txn::POLICIES;
33         class addr_parity_policy extends policy_imp#(addr_p_txn);
34             // ... unchanged from previous example
35         endclass
36
37         static function addr_parity_policy FIXED_PARITY(bit value);
38             FIXED_PARITY = new(value);
39         endfunction
40     endclass: POLICIES
41 endclass
42
43 class addr_constrained_txn extends addr_p_txn;
44     function new;
45         policy_queue pcy;
46
47         addr_constrained_txn::POLICIES::addr_permit_policy permit = new();
48         addr_constrained_txn::POLICIES::addr_prohibit_policy prohibit = new();
49
50         permit.add('h00000000, 'h0000FFFF);
51         permit.add('h10000000, 'h1FFFFFFF);
52         pcy.push_back(permit);
53
54         prohibit.add('h13000000, 'h130FFFFF);
55         pcy.push_back(prohibit);
56
57         pcy.push_back(addr_constrained_txn::POLICIES::FIXED_PARITY(1'b1));
58
59         this.policy = {pcy};
60     endfunction
61 endclass

```

Fig. 11: Address transaction classes with embedded policies

In the above code, there were few changes from the previous iteration. All the policy classes were migrated into an embedded



**POLICIES** class, and a static constructor was created for the parity policy to simplify its creation. The parity **POLICIES** class was set as a child of the initial **POLICIES** class, which further simplified the usage of the classes in the constraint block. Finally, policies in the constraint block are accessed through the **addr\_constrained\_txn::POLICIES::** accessor.

#### IV. PROBLEM #3: BOILERPLATE OVERLOAD

The third problem is that policies are relatively expensive to define since you need at a minimum, a class definition, a constructor, and a constraint. This will be relatively unavoidable for complex policies, such as those defining a relationship between multiple specific class attributes. For generic policies, such as equality constraints, range constraints, or set membership (keyword **inside**) constraints, macros can be used to drastically reduce the expense and risk of defining common policies. The macros are responsible for creating the specialized policy class for the required constraint within the target class, as well as a static constructor function that is used to create new policy instances of the class. Additional macros are utilized for setting up the embedded **POLICIES** class within the target class. These macros can be used hand in hand with the non-macro policy classes needed for complex constraints, if necessary.

```

1 // Create a base embedded POLICIES class
2 `define start_policies(cls) \
3 class POLICIES; \
4 `LOCAL_POLICY_IMP(cls)
5
6 // Create a child embedded POLICIES class
7 `define start_extended_policies(cls, parent) \
8 class POLICIES extends parent::POLICIES; \
9 `LOCAL_POLICY_IMP(cls)
10
11 // End the embedded POLICIES class
12 `define end_policies \
13 endclass: POLICIES
14
15 // Create the base policy type
16 `define LOCAL_POLICY_IMP(cls) \
17     typedef policy_imp#(cls) base_policy;
18
19
20 // Fixed-value policy class and constructor macro
21 `define fixed_policy(POLICY, TYPE, field) \
22 `FIXED_POLICY_CLASS(POLICY, TYPE, field) \
23 `FIXED_POLICY_CONSTRUCTOR(POLICY, TYPE, field)
24
25 `define FIXED_POLICY_CLASS(POLICY, TYPE, field) \
26     class POLICY`_policy extends base_policy; \
27     protected TYPE m_fixed_value; \
28 \
29     constraint c_fixed_value { \
30         m_item != null -> m_item.field == m_fixed_value; \
31     } \
32 \
33     function new(TYPE value); \
34         this.m_fixed_value = value; \
35     endfunction \
36     endclass: POLICY`_policy
37
38 `define FIXED_POLICY_CONSTRUCTOR(POLICY, TYPE, field) \
39     static function POLICY`_policy POLICY(TYPE value); \
40         POLICY = new(value); \

```

Fig. 12: Macros for setting up the embedded **POLICIES** class and a fixed value policy

The **start\_policies**, **start\_extended\_policies**, and **end\_policies** macros replace the embedded **class POLICIES** and **endclass** statements within the constrained class. They set up class inheritance as needed and create a local typedef for the **policy\_imp** parameterized type. The **fixed\_policy** macro wraps two additional macros responsible for creating the policy class and a static constructor for the class. In this example, a policy class that lets you constrain a property to a fixed value is defined.

```

1  class addr_txn extends uvm_object;
2      rand bit [31:0]  addr;
3      rand int         size;
4      rand policy_queue policy;
5
6      constraint c_size {size inside {1, 2, 4};}
7
8      function void pre_randomize;
9          foreach(policy[i]) policy[i].set_item(this);
10     endfunction
11
12     `start_policies(addr_txn)
13     `include "addr_policies.svh"
14     `end_policies
15 endclass
16
17 class addr_p_txn extends addr_txn;
18     rand bit parity;
19
20     constraint c_parity {parity == $countones(addr) % 2;}
21
22     `start_extended_policies(addr_p_txn, addr_txn)
23     `fixed_policy(FIXED_PARITY, bit, parity)
24     `end_policies
25 endclass
26
27 class addr_constrained_txn extends addr_p_txn;
28     // ... unchanged from previous example
29 endclass

```

Fig. 13: Simplified address transaction classes using the policy macros

The base **addr\_txn** class has complex policies with a relationship between the **addr** and **size** fields, so rather than creating a policy macro that will only be used once, they can either be left as-is within the embedded policies class, or moved to a separate file and included with **`include** as was done here to keep the transaction class simple. The child parity transaction class is able to use the **`fixed\_policy** macro to constrain the **parity** field. The constraint block remains the same as the previous example.

#### V. PROBLEM #4: UNEXPECTED POLICY REUSE BEHAVIOR AND OPTIMIZING FOR LIGHTWEIGHT POLICIES

A fourth problem we noticed in our initial usage of policy classes was occasional unexpected behavior when attempting to reuse a policy. It is hard to quantify and we never spent time to create a reproducible test case, but some of the policies behaved like they “remembered” being randomized when used repeatedly and wouldn’t reapply their constraints during subsequent **randomize** calls. Rather than spending time to find an exact diagnosis of the issue, we were focused on transitioning our policy approach to a “use once and discard” approach, where we would treat a policy as disposable and model them as extremely lightweight classes. This change happened to solve the randomization issue as well.

We expected that we would need hundreds or thousands of policies per test, so we aimed to minimize the memory requirements and streamline functionality as much as possible. To minimize the footprint of policies, we chose to implement them as simple classes rather than extensions of `uvm_object`. We duplicated some useful capabilities such as a basic copy routine, while omitting more expensive functionality such as factory compatibility or object comparison. The buggy randomization behavior implied we needed to pass in fresh policy instances each time we randomized an object. Two features helped to accomplish this: static constructor functions and a `copy` method.

```
1 static function addr_parity_policy FIXED_PARITY(bit value);
2     FIXED_PARITY = new(value);
3 endfunction
4 ...
5 pcy.push_back(addr_constrained_txn::POLICIES::FIXED_PARITY(1'b1));
```

Fig. 14: Example static constructor function from the `FIXED_PARITY` policy class

Using static functions to create new instances of the policies provides an easy way to create and use a policy with a single function call. The policy is created and immediately pushed onto the policy queue, then discarded and recreated the next time it is needed.

```
1 interface class policy;
2     ...
3     pure virtual function policy copy();
4 endclass: policy
```

Fig. 15: Example static constructor function from the `FIXED_PARITY` policy class

Adding a `copy` method to the policy classes provides a way for a policy to return a fresh copy of itself with the same configuration. This is useful when a policy needs to be reused multiple times. The policy can be copied and pushed onto the policy queue multiple times, or onto other policy queues, and each copy will constrain the field separately. The `copy` method is `pure virtual`, so it must be implemented by each policy class.

After migrating to this pattern of reuse, the issues we were seeing with unapplied constraints disappeared.

## VI. MORE IMPROVEMENTS TO THE POLICY PACKAGE

The examples presented so far are functional, but are lacking many features that would be useful in a real-world implementation. The following examples will present additional improvements to the policy package that will make it more practical efficient to use.

### A. Expanding the `policy` interface class

The following `policy` interface class adds additional methods for managing a policy.

```
1 interface class policy;
2
3     pure virtual function string name();
4
5     pure virtual function string type_name();
6
7     pure virtual function string description();
8
9     pure virtual function bit item_is_compatible(uvm_object item);
10
11     pure virtual function void set_item(uvm_object item);
12
13     pure virtual function policy copy();
```

```

14
15   endclass: policy

```

Fig. 16: Expanded **policy** interface class

The **name**, **description**, and **copy** methods are implemented by the policy (or policy macro) directly and provide reporting information useful when printing messages about the policy to the log for the former two, or specific behavior for making a copy for the latter. The remaining three methods are implemented by **policy\_imp** and are shared by all policies.

#### B. Better type safety checking and reporting in **policy\_imp** methods

Some of the benefits of above methods can be seen by examining the new **set\_item** method used by **policy\_imp**.

```

1  virtual function void set_item(uvm_object item);
2      if (item == null) begin
3          `uvm_error("policy::set_item()", "NULL item passed")
4
5      end else if ((this.item_is_compatible(item)) && $cast(this.m_item, item)) begin
6          `uvm_info(
7              "policy::set_item()",
8              $sformatf(
9                  "policy <%s> applied to item <%s>: %s",
10                 this.name(), item.get_name(), this.description()
11             ),
12             UVM_FULL
13         )
14         this.m_item.rand_mode( 1 );
15
16     end else begin
17         `uvm_warning(
18             "policy::set_item()",
19             $sformatf(
20                 "Item <%s> type <%s> is not compatible with policy <%s> type <%s>",
21                 item.get_name(), item.get_type_name(), this.name(), this.type_name()
22             )
23         )
24         this.m_item = null;
25         this.m_item.rand_mode( 0 );
26     end
27 endfunction: set_item

```

Fig. 17: **set\_item** method from **policy\_imp**

The **set\_item** method makes use of all the reporting methods to provide detailed log messages when **set\_item** succeeds or fails. Additionally, the **item\_is\_compatible** is used before the **\$cast** method is called and the **rand\_mode** state is kept consistent with the result of the cast.

#### C. Replacing **policy\_list** with **policy\_queue** and **policy\_container**

Eagle-eyed readers might have noticed the lack of presence of a **policy\_list** class in any of the examples above after migrating to the improved policy interface. Rather, a single typedef is all that is necessary to manage policies in a class.

```

1  typedef policy policy_queue[$];

```

Fig. 18

The **policy\_queue** type is capable of storing any policy that implements the **policy** interface. The default queue methods are sufficient for basic usage, but by introducing another interface class, the direct management of the queue can be abstracted away from the user.

```
1 interface class policy_container;
2
3     // Queries
4     pure virtual function bit has_policies();
5
6     // Assignments
7     pure virtual function void set_policies(policy_queue policies);
8
9     pure virtual function void add_policies(policy_queue policies);
10
11    pure virtual function void clear_policies();
12
13    // Access
14    pure virtual function policy_queue get_policies();
15
16    // Copy
17    pure virtual function policy_queue copy_policies();
18
19 endclass: policy_container
```

Fig. 19: The **policy\_container** interface class

This interface class should be implemented by any class that needs to manage a policy queue. The method implementations can then be used to manipulate a policy queue as needed, including adding to, replacing, or clearing the queue, and returning a handle to the queue or a copy of the queue to use elsewhere.

#### D. Creating a **policy\_container** base class implementation

The **policy\_container** interface class allows for a base class implementation that can be used to manage constraints across the entire class hierarchy.

```
1 class policy_object #(type BASE=uvm_object) extends BASE implements
2     policy_container;
3
4     protected policy_queue m_policies;
5
6     // Queries
7     virtual function bit has_policies();
8         return( this.m_policies.size() > 0 );
9     endfunction: has_policies
10
11    // Assignments
12    virtual function void set_policies(policy_queue policies);
13        if( this.has_policies() ) begin
14            `uvm_warning( "policy", "set_policies() replacing existing policies" )
15        end
16        this.m_policies = {};
17        foreach( policies(i) ) try_add_policy( policies[i] );
18    endfunction: set_policies
```

```

19     virtual function void add_policies(policy_queue policies);
20         foreach( policies[i] ) try_add_policy( policies[i] );
21     endfunction: add_policies
22
23     virtual function void clear_policies();
24         if ( this.has_policies() ) begin
25             `uvm_info( "policy", $sformatf("clearing [%0d] policies from %s",
26             this.m_policies.size(), this.get_name()), UVM_FULL)
27         end
28         this.m_policies = {};
29     endfunction: clear_policies
30
31     // Access
32     virtual function policy_queue get_policies();
33         return( this.m_policies );
34     endfunction: get_policies
35
36     // Copy
37     virtual function policy_queue copy_policies();
38         copy_policies = {};
39         foreach( this.m_policies[i] ) begin
40             copy_policies.push_back( this.m_policies[i].copy );
41         end
42     endfunction: copy_policies
43
44     protected function void try_add_policy( policy new_policy );
45         if (new_policy.item_is_compatible(this)) begin
46             `uvm_info( "policy", $sformatf("adding policy %s to %s",
47             new_policy.name, this.get_name()), UVM_FULL)
48             new_policy.set_item( this );
49             this.m_policies.push_back( new_policy );
50         end else begin
51             `uvm_warning( "policy", $sformatf("policy %s not compatible with target
52             %s", new_policy.name, this.get_name()))
53         end
54     endfunction: try_add_policy
55 endclass: policy_object

```

Fig. 20: A **policy\_object** base class implementation

This base class implements all the **policy\_container** functionality. Treating it as a mixin allows you to apply it to different categories of transactions.

```

1 class base_txn extends policy_object #(uvm_sequence_item);
2
3 class base_seq #(type REQ=uvm_sequence_item, RSP=REQ) extends policy_object #(
4     uvm_sequence #(REQ, RSP) );
5
6 class cfg_object extends policy_object #(uvm_object);

```

Fig. 21: Example classes using the **policy\_object** base class

Each of these classes will contain a separate policy queue, but each will share an API and can be used in the same way. The **policy\_object** base class can be used as a mixin to any class that needs to manage a policy queue.

### *E. Protecting the policy queue*

A subtle but significant additional benefit to using a base `policy_object` class along with the `policy_container` API is the ability to mark the container's policy queue as protected and prevent direct access to it.

The original implementation called `set_item` on each policy during the `pre_randomize` stage. That was necessary because the `policy_queue` was public, so there was nothing to prevent callers from adding policies without linking them to the target class.

Using an interface class and making the implementation private means the callers may only set or add policies using the available interface class routines, and because those routines are solely responsible for applying policies, they can also check compatibility (and filter incompatible policies) and call `set_item` immediately. This can be seen in the example `policy_object` implementation above, in the usage of the protected function `try_add_policy`.

By calling `set_item` when the policy is added to the queue, all policies will be associated with the target item automatically, so there is no need to do it during `pre_randomize`. This eliminates an easily-overlooked requirement for classes extending `policy_object` to make sure they call `super.pre_randomize()` in their local `pre_randomize` function.

## VII. CONCLUSION

The improvements to the policy package presented in this paper provide a more robust and efficient implementation of policy-based constraints for SystemVerilog. The policy package is now capable of managing constraints across an entire class hierarchy, and the policy definitions are tightly paired with the class they constrain. The use of macros reduces the expense of defining common policies, while still allowing great flexibility in any custom policies necessary.

The complete policy package is available in Appendix A, with additional macro examples available in Appendix B. A functional package is also available for download [6] which can be included directly in a project to start using policies immediately.

## REFERENCES

- [1] J. Dickol, "SystemVerilog Constraint Layering via Reusable Randomization Policy Classes," DVCon, 2015.
- [2] J. Dickol, "Complex Constraints: Unleashing the Power of the VCS Constraint Solver," SNUG Austin, September 29, 2016.
- [3] K. Vasconcellos, J. McNeal, "Configuration Conundrum: Managing Test Configuration with a Bite-Sized Solution," DVCon, 2021.
- [4] C. McClish, "Bi-Directional UVM Agents and Complex Stimulus Generation for UDN and UPF Pins," DVCon, 2021.
- [5] J. Vance, J. Montesano, M. Litterick, J. Sprott, "Be a Sequence Pro to Avoid Bad Con Sequences," DVCon, 2019.
- [6] D. Mills, C. Haldane, "policy\_pkg Source Code," [https://github.com/DillanCMills/policy\\_pkg](https://github.com/DillanCMills/policy_pkg).

## APPENDIX A POLICY PACKAGE

### A. *policy\_pkg.sv*

```
1  `ifndef __POLICY_PKG__
2  `define __POLICY_PKG__
3
4  `include "uvm_macros.svh"
5
6  package policy_pkg;
7
8      import uvm_pkg::*;
9
10     `include "policy.svh"
11
12     typedef policy POLICY_QUEUE[$];
13
14     `include "policy_container.svh"
15
16     `include "policy_imp.svh"
17     `include "policy_list.svh"
18
19 endpackage: policy_pkg
20
21 `endif
```

### B. *policy.svh*

```
1  `ifndef __POLICY__
2  `define __POLICY__
3
4  interface class policy;
5
6      pure virtual function string name();
7
8      pure virtual function string type_name();
9
10     pure virtual function string description();
11
12     pure virtual function bit item_is_compatible(uvm_object item);
13
14     pure virtual function void set_item(uvm_object item);
15
16     pure virtual function policy copy();
17
18 endclass: policy
19
20 `endif
```

### C. *policy\_container.svh*



```

1  `ifndef __POLICY_CONTAINER__
2  `define __POLICY_CONTAINER__
3
4  interface class policy_container;
5
6      // Queries
7      pure virtual function bit has_policies();
8
9      // Assignments
10     pure virtual function void set_policies(policy_queue policies);
11
12     pure virtual function void add_policies(policy_queue policies);
13
14     pure virtual function void clear_policies();
15
16     // Access
17     pure virtual function policy_queue get_policies();
18
19     // Copy
20     pure virtual function policy_queue copy_policies();
21
22 endclass: policy_container
23
24 `endif

```

#### D. policy\_imp.svh

```

1  `ifndef __POLICY_IMP__
2  `define __POLICY_IMP__
3
4  virtual class policy_imp #(type ITEM=uvm_object) implements policy;
5
6      protected rand ITEM m_item;
7
8      pure virtual function string name();
9      pure virtual function string description();
10     pure virtual function policy copy();
11
12     virtual function string type_name();
13         return( ITEM::type_name );
14     endfunction: type_name
15
16     virtual function bit item_is_compatible(uvm_object item);
17         ITEM local_item;
18
19         if (item == null) return( 0 );
20         else return( $cast(local_item, item) );
21     endfunction: item_is_compatible
22
23     virtual function void set_item(uvm_object item);
24         if (item == null) begin
25             `uvm_error("policy::set_item()", "NULL item passed")

```

```

26
27     end else if ((this.item_is_compatible(item)) && $cast(this.m_item, item))
28     begin
29         `uvm_info(
30             "policy::set_item()",
31             $sformatf(
32                 "policy <%s> applied to item <%s>: %s",
33                 this.name(), item.get_name(), this.description()
34             ),
35             UVM_FULL
36         )
37         this.m_item.rand_mode( 1 );
38     end else begin
39         `uvm_warning(
40             "policy::set_item()",
41             $sformatf(
42                 "Item <%s> type <%s> is not compatible with policy <%s> type
43         <%s>",
44                 item.get_name(), item.get_type_name(), this.name(),
45         this.type_name()
46             )
47         )
48         this.m_item = null;
49         this.m_item.rand_mode( 0 );
50     end
51     endfunction: set_item
52
53 endclass: policy_imp
`endif

```

## APPENDIX B POLICY MACROS

### A. *policy\_macros.svh*

```
1  `ifndef __POLICY_MACROS__
2  `define __POLICY_MACROS__
3
4  //=====
5  // Embedded POLICIES class macros
6  //=====
7
8  `define start_policies(cls)          \
9  class POLICIES;                      \
10 `m_base_policy(cls)
11
12 `define start_extended_policies(cls, parent) \
13 class POLICIES extends parent::POLICIES;    \
14 `m_base_policy(cls)
15
16 `define end_policies                  \
17 endclass: POLICIES
18
19 `define m_base_policy(cls)            \
20     typedef policy_imp#(cls) base_policy;
21
22
23 //=====
24 // Policy template macros
25 //=====
26
27 `include "constant_policy.svh"
28 `include "fixed_policy.svh"
29 `include "ranged_policy.svh"
30 `include "set_policy.svh"
31
32 `endif
```

### B. *constant\_policy.svh*

```
1  `ifndef __CONSTANT_POLICY__
2  `define __CONSTANT_POLICY__
3
4  // Full policy definition
5  `define constant_policy(POLICY, TYPE, field, constant) \
6  `CONST_POLICY_CLASS(POLICY, TYPE, field, constant)    \
7  `CONST_POLICY_CONSTRUCTOR(POLICY)
8
9  // Policy class definition
10 `define CONST_POLICY_CLASS(POLICY, TYPE, field, constant) \
11     class POLICY`_policy extends base_policy              \
12         typedef TYPE l_field_t;                             \
13
14     virtual function string name();                          \
```

```

15         return (`"POLICY`");
16     endfunction: name
17
18     virtual function string description();
19         return (`"(field==constant)`");
20     endfunction: description
21
22     virtual function policy copy();
23         copy = new();
24     endfunction: copy
25
26     constraint c_const_value {
27         (item != null) -> (item.field == l_field_t'(constant));
28     }
29
30     function new();
31     endfunction: new
32 endclass: POLICY``_policy
33
34 // Policy constructor definition
35 `define CONST_POLICY_CONSTRUCTOR(POLICY)
36     static function POLICY``_policy POLICY();
37         POLICY = new();
38     endfunction: POLICY
39
40 `endif

```

### C. fixed\_policy.svh

```

1  `ifndef __FIXED_POLICY__
2  `define __FIXED_POLICY__
3
4  // Full policy definition
5  `define fixed_policy(POLICY, TYPE, field, RADIX="%0p")
6  `FIXED_POLICY_CLASS(POLICY, TYPE, field, RADIX)
7  `FIXED_POLICY_CONSTRUCTOR(POLICY, TYPE, RADIX)
8
9  // Policy class definition
10 `define FIXED_POLICY_CLASS(POLICY, TYPE, field, RADIX="%0p")
11     class POLICY``_policy extends base_policy
12         typedef TYPE      l_field_t;
13
14         protected TYPE      m_fixed_value;
15         protected string    m_radix=RADIX;
16
17         constraint c_fixed_value {
18             (item != null) -> (item.field == l_field_t'(m_fixed_value));
19         }
20
21         function new(TYPE value, string radix=RADIX);
22             this.set_value(value);
23             this.set_radix(radix);
24         endfunction: new
25

```

```

26     virtual function string name();
27         return ({
28             ~"POLICY(field==",
29             $sformatf(m_radix, m_fixed_value),
30             ~")~"
31         });
32     endfunction: name
33
34     virtual function void set_value(TYPE value);
35         this.m_fixed_value = value;
36     endfunction: set_value
37
38     virtual function TYPE get_value();
39         return (this.m_fixed_value);
40     endfunction: get_value
41
42     virtual function void set_radix(string radix);
43         this.m_radix = radix;
44     endfunction: set_radix
45
46     virtual function string get_radix();
47         return (this.m_radix);
48     endfunction: get_radix
49 endclass: POLICY`*_policy
50
51 // Policy constructor definition
52 `define FIXED_POLICY_CONSTRUCTOR(POLICY, TYPE, RADIX="%0p")
53     static function POLICY`*_policy POLICY(TYPE value, string radix=RADIX);
54         POLICY = new(value, radix);
55     endfunction: POLICY
56
57 `endif

```

#### D. ranged\_policy.svh

```

1  `ifndef __RANGED_POLICY__
2  `define __RANGED_POLICY__
3
4  // Full policy definition
5  `define ranged_policy(POLICY, TYPE, field, RADIX="%0p")
6  `RANGED_POLICY_CLASS(POLICY, TYPE, field, RADIX)
7  `RANGED_POLICY_CONSTRUCTOR(POLICY, TYPE, RADIX)
8
9  // Policy class definition
10 `define RANGED_POLICY_CLASS(POLICY, TYPE, field, RADIX="%0p")
11     class POLICY`*_policy extends base_policy
12         typedef TYPE        l_field_t;
13
14         protected TYPE      m_low;
15         protected TYPE      m_high;
16         protected bit       m_inside;
17         protected string    m_radix=RADIX;
18
19         constaint c_ranged_value {

```

```

20         (item != null) ->
21         ((!m_inside) ^ (item >= m_low && item <= m_high));
22     }
23
24     function new(
25         TYPE    low,
26         TYPE    high=low,
27         bit     inside=1'b1,
28         string  radix=RADIX
29     );
30         this.set_range(low, high);
31         this.set_inside(inside);
32         this.set_radix(radix);
33     endfunction: new
34
35     virtual function string name();
36         return ({
37             ~"POLICY(field ",
38             m_inside ? "inside [" : "outside [",
39             $sformatf(m_radix, m_low),
40             ", ",
41             $sformatf(m_radix, m_high),
42             ~"])"
43         });
44     endfunction: name
45
46     virtual function void set_range(TYPE low, TYPE high);
47         if (low <= high) begin
48             this.m_low = low;
49             this.m_high = high;
50         end else begin
51             this.m_low = high;
52             this.m_high = low;
53         end
54     endfunction: set_range
55
56     virtual function TYPE get_low();
57         return (this.m_low);
58     endfunction: get_low
59
60     virtual function TYPE get_high();
61         return (this.m_high);
62     endfunction: get_high
63
64     virtual function void set_inside(bit inside);
65         this.m_inside = inside;
66     endfunction: set_inside
67
68     virtual function bit get_inside();
69         return (this.m_inside);
70     endfunction: get_inside
71
72     virtual function void set_radix(string radix);
73         this.m_radix = radix;

```

```

74         endfunction: set_radix                                     \
75                                                                 \
76         virtual function string get_radix();                     \
77             return (this.m_radix);                               \
78         endfunction: get_radix                                    \
79     endclass: POLICY`__policy                                     \
80                                                                 \
81 // Policy constructor definition                                  \
82 `define RANGED_POLICY_CONSTRUCTOR(POLICY, TYPE, RADIX="%0p")    \
83     static function POLICY`__policy POLICY(                      \
84         TYPE    low,                                             \
85         TYPE    high=low,                                        \
86         bit     inside=1'b1,                                     \
87         string  radix=RADIX                                     \
88     );                                                           \
89         POLICY = new(low, high, inside, radix);                  \
90     endfunction: POLICY                                          \
91                                                                 \
92 `endif

```

#### E. set\_policy.svh

```

1  `ifndef __SET_POLICY__
2  `define __SET_POLICY__
3
4  // Full policy definition
5  `define set_policy(POLICY, TYPE, field, RADIX="%0p")           \
6  `SET_POLICY_CLASS(POLICY, TYPE, field, RADIX)                 \
7  `SET_POLICY_CONSTRUCTOR(POLICY, TYPE, RADIX)                  \
8
9  // Policy class definition
10 `define SET_POLICY_CLASS(POLICY, TYPE, field, RADIX="%0p")    \
11     class POLICY`__policy extends base_policy                 \
12         typedef TYPE      l_field_array_t[];                  \
13                                                                 \
14         protected l_field_array_t  m_values;                  \
15         protected bit               m_inside;                 \
16         protected string            m_radix=RADIX;             \
17                                                                 \
18         constraint c_set_value {                                \
19             (item != null) ->                                  \
20                 ((m_inside) ^ (item.field inside {m_values})); \
21         }                                                        \
22                                                                 \
23         function new(                                           \
24             l_field_array_t values,                             \
25             bit             inside=1'b1,                        \
26             string          radix=RADIX                         \
27         );                                                       \
28             this.set_values(values);                            \
29             this.set_inside(inside);                           \
30             this.set_radix(radix);                             \
31         endfunction: new                                        \
32

```

```

33     virtual function string name();
34         string values_str = "";
35
36         foreach(m_values[i])
37             values_str = {
38                 values_str,
39                 $sformatf(m_radix, m_values[i]),
40                 i == m_values.size()-1 ? "" : ", ";
41
42         return ({
43             ~"POLICY(field ",
44             m_inside ? "inside {" : "outside {" ,
45             values_str,
46             ~"}~"
47         });
48     endfunction: name
49
50     virtual function void set_values(l_field_array_t values);
51         this.m_values = values;
52     endfunction: set_values
53
54     virtual function l_field_array_t get_values();
55         l_field_array_t l_array;
56         foreach (this.m_values[i])
57             l_array[i] = this.m_values[i];
58         return (l_array);
59     endfunction: get_values
60
61     virtual function void set_inside(bit inside);
62         this.m_inside = inside;
63     endfunction: set_inside
64
65     virtual function bit get_inside();
66         return (this.m_inside);
67     endfunction: get_inside
68
69     virtual function void set_radix(string radix);
70         this.m_radix = radix;
71     endfunction: set_radix
72
73     virtual function string get_radix();
74         return (this.m_radix);
75     endfunction: get_radix
76     endclass: POLICY`*_policy
77
78 // Policy constructor definition
79 `define SET_POLICY_CONSTRUCTOR(POLICY, TYPE, RADIX="%0p")
80     typedef TYPE POLICY`*_array_t[];
81     static function POLICY`*_policy POLICY(
82         POLICY`*_array_t values,
83         bit                inside=1'b1,
84         string              radix=RADIX
85     );
86         POLICY = new(values, inside, radix);

```



```
87     endfunction: POLICY
88
89 `endif
```